PSDTOOL - A DMAP ENHANCEMENT TO HARMONIC/RANDOM RESPONSE ANALYSIS IN MSC/NASTRAN

Alwar Parthasarathy
Mohamed Elzeki
Vivianne Abramovici

Advanced Technology Systems Group
SPAR AEROSPACE LIMITED
Toronto, Canada

ABSTRACT

This Paper presents a DMAP framework called **PSDTOOL** that significantly enhances the scope of **harmonic/random response** analysis in MSC/NASTRAN. PSDTOOL addresses the case where grid-point accelerations (and displacements) form the primary output of a dynamic analysis and locates **response-maxima** within and across subcases and frequencies. In the common aerospace application where the input-excitation is specified as a **base-motion PSD-spectrum** in the form of a scalar or a diagonal matrix, PSDTOOL outputs the **RMS** (and RSS) responses at **all the grid-points** of a finite-element model as a **single OLOAD-type table** and locates **peak-RMS** responses over the whole system or over individual user-defined **subsystems**. For facilitating high-level system design and integration, PSDTOOL includes a novel capability for computing **mass-weighted average-responses at 'centers-of-mass'** of system and subsystems. The goal is to automate and condense output-generation and to produce design-data with speed and confidence. PSDTOOL offers some incidental conveniences such as a conventional **relative-motion** formulation of base-excitation analysis and output-screening in a common co-ordinate system. The DMAP scheme is implemented as an ALTER in a **DBLOCATE-restart** from a mode-extraction database and the various special features are driven by non-standard PARAMeters. Extension to element-output is being envisioned.

1.0 INTRODUCTION

Certain user-oriented enhancements to the scope of **harmonic and random response analysis** in MSC/NASTRAN [1] are the focus of the present work. Of immediate interest are the grid-point accelerations (and displacements) in a structure subjected to harmonic/random excitation and analyzed via the normal-mode method. The enhancements are viewed as a valuable tool in the design process and are implemented via a DMAP-ALTER framework called **PSDTOOL.**

In **harmonic-response** analysis, the standard XYPRINT request of MSC/NASTRAN helps identify SORT2-maxima of **selected components** of response over the range of output-frequencies and within each of a set of subcases. However, for speed and accuracy in assessing practical models of relatively large size, there exists a need for a more comprehensive and automated **SORT1/SORT2 search-facility** that would scan/filter output over **grid-points** and frequencies and **across** and within subcases.

Standard MSC/NASTRAN dictates that **random-response** results such as power-spectral-density (PSD) and root-mean-square (RMS) values be extractable only as XYOUT-output. Thus, for obtaining **any PSD-type results,** the user needs to select and specify **individual components** of response at desired nodes and elements in the XYOUT-request. This process would tend to be tedious and incomplete if one needs to identify locations of maximum response in a relatively large model. The SORT1/SORT2 search-capability mentioned above may indeed be invoked to enhance the process by identifying **maxima of PSD**-values. What would be of even greater appeal is a facility that would compute a single **SORT1**-table containing the **RMS-responses at all the grid-points** of the structure and that would also identify **peak-RMS** values.

Engineers involved in system-integration often focus on **average-responses** at '**centers-of-mass**' (CM) of system and **subsystems**, rather than on the detailed responses at all the myriad grid-points of a finite-element (FE) model. Indeed, the CM may not even be a true grid-point in the FE-model. A capability to compute the **mass-weighted** mean-responses of the CMs of **system and user-partitioned subsystems** would be of practical interest.

A common form of excitation in aerospace applications is a **uniform base-acceleration.** In standard MSC/NASTRAN, such input is accommodated via the **Large-Mass** approach. Though general in scope, this method is somewhat awkward and prone to misuse. An implementation of the conventional **Relative-Motion** approach [2] as an alternative may be welcome.

PSDTOOL addresses the above needs within certain broad limits.

2.0 ASSUMPTIONS

It is assumed that the entire model constitutes the Residual (superelement) Structure and that Scalar and Extra Points are absent. However, the user may define convenient partitions (i.e. subsystems) of the structure, via special DMIG-cards, for output-scanning on a per-subsystem basis. The only output-items that can be searched for maxima are the displacements and accelerations at grid-points. If the excitation is random, the frequency-dependent input-PSD is restricted to be either a scalar function or a diagonal-matrix signifying uncorrelated loading conditions.

3.0 PROCEDURAL OVERVIEW

At present, PSDTOOL is implemented as part of a **two-phase** analysis in Version 66A of MSC/NASTRAN [3]. Phase-I is a **standard SOL-63** (cold-start) execution for determining the natural frequencies and modes of the structure. The database containing mode-extraction results is saved. Phase-II is a **non-standard SOL-71 Restart** (Modal Frequency and Random Response) that includes the PSDTOOL ALTER-packet. The restart is of the DBLOCATE-type, reading modal results from the Phase-I database without corrupting it. The present Paper would devote most of its attention to Phase-II.

3.1 HARMONIC RESPONSE

With PARAM,DDRMM,-1 as mandatory input, a full UGVS matrix of harmonic (i.e., frequency-response) displacements is created. The matrix has 'NG' rows and 'NOLB1 x NSUBC' columns, where 'NG' is the size of the 'G-Set', 'NOLB1' is the number of excitation/response frequencies and 'NSUBC' is the number of load-subcases. Only the magnitudes of the terms of the (complex) matrix UGVS are of interest in the present work.

Displacements and accelerations have the simple relationship given by,

$$\text{'ACCEL'} = (-\Omega^2) \,.\, \text{'DISP'} \qquad (3.1.1)$$

where Ω is the circular frequency of response. In a given run, the user chooses the desired type of response via a parameter called 'DISACC'.

The UGVS-matrix could be SORT1-searched (row-searched) and/or SORT2-searched (column-searched) for maxima of amplitudes **via matrix-operations.**

Since the solution-frequencies are required in matrix-operations, the standard OLB1-table is transformed into matrix-form via a pair of **APPEND-loops**.

3.2 PSD-ANALYSIS VIA MASKED-SINE

The standard input-output PSD-relationship is, from equations (6) and (8) of Section 12.2 of Ref.[1],

$$S_{out} = H \cdot S_{in} \cdot H^* \qquad (3.2.1)$$

where, 'S' the spectral-density and 'H' the frequency-response are functions of the excitation-frequency and where the superscript '*' denotes 'conjugate-transpose'. Standard MSC/NASTRAN calculates only the diagonal terms of S_{out} (i.e., auto-spectral-densities) and PSDTOOL conforms to this practice.

When S_{in} is a scalar, we may write the **auto**-spectral-densities of grid-point responses as a UGVS-like matrix S_{out} given by,

$$S_{out} = (H \cdot R_{in}) : (R_{in} \cdot H^*) \qquad (3.2.2)$$

where R_{in} is the **square-root** of S_{in} and ':' denotes term-by-term multiplication.

If S_{in} is an NxN diagonal matrix representing a set of 'N' uncorrelated input-conditions, S_{out} is computed as the **sum** of 'N' product-matrices similar to the one shown on the right-hand side of equation (3.2.1).

Thus, for the particular cases of PSD-excitation considered here, the PSD-values of excitation and response may be viewed as **'masked' harmonic** quantities. Instead of inputing S_{in} via the standard RANDPS and TABRND cards, we would input R_{in} via the RLOADi and TABLEDi cards. PSDTOOL would generate an S_{out}-matrix of size 'NG x NOLB1', via explicit manipulation of the terms of the UGVS-matrix. Matrix S_{out} could be searched for maxima (and XYPLOTed) in the same manner as UGVS.

An 'NG x 1' matrix, Q_{out}, of RMS-values of random-response could be determined by trapezoidal integration of S_{out} over the frequency-range, as given below (equation (11) of the preceding reference) :-

$$Q_{out} = [(1/(4\pi)) \cdot \sum\{(S_{i+1} + S_i)\ (\Omega_{i+1} - \Omega_i)\}]^{1/2} \qquad (3.2.3)$$

where S_i is the value of S_{out} at Ω_i; and, the summation is for 'i' in the range $1 \leq i \leq$ (NOLB1-1).

We rewrite Q_{out} in matrix-form as

$$Q_{out} = [(0.5) \cdot S_{out}^T \cdot \text{FQDEL}]^{1/2} \qquad (3.2.4)$$

where 'FQDEL' is a vector of length 'NOLB1', made up of the following sequence of frequency-differences in Hertz units :-

$\{ (f_2-f_1),\ (f_3-f_1),\ (f_4-f_2),\ (f_5-f_3),\ \ldots,\ (f_{NOLB1} - f_{NOLB1-2}),\ (f_{NOLB1} - f_{NOLB1-1}) \}$

As an incidental convenience in assessing response-levels, a matrix Z_{out} that smooths out directional bias is also computed. For any grid-point 'j' in Z_{out}, all DOF's except the first are set to zero. The first component, $z_1(j)$, is calculated as the 'square-root of the sum of the squares' (RSS) of the three translational components of grid-point 'j' in Q_{out}. Thus,

$$z_1(j) \quad = \quad [q_1^2(j) + q_2^2(j) + q_3^2(j)]^{1/2} \qquad (3.2.5)$$

Q_{out} and Z_{out} may be searched for peaks of RMS and RSS respectively.

3.3 SUBSYSTEM DEFINITION

Subsystems or Sections of the structure may be defined at the discretion of the user. They are useful in the output-scanning phase, for the identification of local response-maxima and local response-averages in different parts of the structure. Subsystems are set up by the user by the assignment of selected grid-points to different groups. A DMIG-matrix called SECTION is input, with each column representing a specific subsystem. Only DOF=1 of any selected grid-point need be entered on the DMIG-card input, with a coefficient of 1.0. PSDTOOL would expand the matrix to all six DOF's. Not all the grid-points in the model need necessarily be assigned. For large models, the DMIG cards may be generated via a FORTRAN-program. If no DMIG-SECTION is input, the whole structure is treated as a single subsystem.

3.4 CENTER-OF-MASS MOTION

If the average-response of a subsystem or the structure is required, the user sets up additional grid-points in the model, one for each subsystem. Though these grid-points are called 'centers' or 'centers-of-mass', they are, at present, only convenient reference-points similar to the standard GRDPNT-point. Their locations are specified by the user and may be chosen to coincide with user-determined mass-centers. A sparse DMIG-matrix called DRUGSEC is input by the user to associate a center-grid with a subsystem. The mass-weighted mean-motion of the center is determined from the motions of the individual grid-points of the subsystem as follows :-

Let 'D' be the '(NSx6) x 6' matrix of rigid-body motions of the 'NS' grid-points of the subsystem due to six unit-motions at the center. This matrix is computed by a call to the VECPLOT-module followed by matrix-manipulations. Let MSS and ACCSEC be respectively the mass-matrix and (complex) acceleration-matrix of the subsystem. We define the 6x6 matrix MR as,

$$MR \quad = \quad D^T \cdot MSS \cdot D \qquad (3.4.1)$$

The mean-accelerations ACCR at the center are computed as,

$$ACCR = (MR)^{-1} . D^T . MSS . ACCSEC \quad (3.4.2)$$

If a fictitious large-mass is present in the model for facilitating base-motion analysis, it is excluded during this calculation. Provision is also made for possible null columns in MR. ACCR is computed in a DO-loop over the set of subsystems. ACCR and the corresponding center-displacements are merged into the overall solution for subsequent output-processing.

3.5 RELATIVE-MOTION APPROACH

In MSC/NASTRAN, the Large-Mass Approach (LMA) for uniform base-acceleration excitation is usually implemented via (a) an optional RBE2 to reduce the base-points to a single node, (b) a CONM2 to add a suitably chosen large mass and, perhaps, large inertias to the base-node and (c) an optional SUPORT. SPC's are excluded at the base-points. Eigensolution is carried out for the modified system and zero-frequency modes are present.

While being fully compatible with the use of the above LMA, PSDTOOL also offers the alternative Relative Motion Approach (RMA) discussed in Ref.[2]. In RMA, the base-points of the structure are SPC'd in the conventional way. There is no need for a fictitious CONM2 (or RBE2 or SUPORT). Eigensolution is standard and there are no special zero-frequency modes. Base-SPCFORCEs are available as output. The RMA-capability of PSDTOOL may be used by itself in what could otherwise be a standard execution. Or it may be used in conjunction with the other non-standard features of PSDTOOL that are limited to processing grid-point output.

In conventional RMA, the (total) displacement-vector 'u' is expressed as the following sum :-

$$u = v + (D . U_b) \quad (3.5.1)$$

where, 'v' is a vector of 'relative' displacements, 'D' the rigid-body matrix defined in Section (3.4) and 'U_b' the 6-DOF displacement-vector at a grid-point identified as the base-point.

The effective dynamic load-vector entering the equations governing 'v' is calculated as,

$$P_{eff} = - (M . D . \ddot{U}_b) \quad (3.5.2)$$

where, 'M' is the mass-matrix, and '$\ddot{U}_b$' the input base-acceleration.

In PSDTOOL, the base is referenced by a disjoint grid-point, the DOF's of which are listed as the U1-Set. Once the relative motions are solved for, total motions, if required, may be obtained from eqn.(3.5.1). Because of the fuller load-vector of RMA compared to that of LMA, run-times and storage-requirements for RMA may be inferior to those for LMA.

3.6 PEAK-SEARCH AND PRINTING

The search for response-peaks is implemented via the **filtering**-option available with the MATMOD module. The normalized filter-value ($0.0 \leq$ filter ≤ 1.0) is an input-parameter and it may be set high enough to trap the maximum. An optional facility is provided for transforming grid-point output to a common co-ordinate system prior to the search.

In **harmonic** response, amplitudes of grid-point response are searched by means of a **nested triple-loop**. The triple-loop is composed of an outer loop over subcases, a middle loop over the six components of response and an inner loop for the two search-types (SORT1/SORT2). The search may be over the entire model or over a user-listed subset of grid-points. Also, two optional 'DMI' input-matrices, called SELSUB and SELDOF, may be employed to restrict the search to certain subcases and certain components respectively. The outer loop may consider one subcase at a time or all the subcases at once. In the middle loop, each component is searched independently of the others. Either or both search-types of the inner loop may be executed in a given run. At present, printing of the results is done via the MATGPR module.

As stated before, the **auto-PSD's** of **random** response are viewed in this work simply as **'masked' harmonic** quantities. Hence, the process of searching for PSD-peaks is similar to that described in the preceding paragraph. The peak-search of '**RMS**' and '**RSS**' quantities of **random** response (Q_{out} and Z_{out} of Section 3.2 respectively) is done for each subsystem separately, in a loop over subsystems. Printing of these quantities (before and/or after peak-search) is done through the friendly **OLOAD** case-control request, unless the user opts for the MATGPR module.

In regard to the handling of multiple subcases, there exists, at present, an incidental procedural difference between harmonic analysis and random analysis. In the former, a subcase-summation facility implying simultaneous application of all the subcase-loadings is yet to be incorporated. In contrast, **random** analysis provides results **only for the 'summed case'**, i.e., the case where the different subcase-loadings represent simultaneous uncorrelated excitations.

3.7 NOTES ON DMAP-ALTER

Figure 1 presents a flow-chart of the DMAP-ALTER driving Phase-II. Tables 1-5 highlight the coding associated with some key boxes of the flow-chart. The error-avoidance step of box-2 is necessary to effect a proper DBLOCATE-restart in Version-66A. Table 1 presents the DMAP for generating RMA effective-loading (box-5), on the basis of eqn.(3.5.2). Table 2 depicts the operation of decoding OLB1 into matrix-form (box-7). Table 3 shows DO-looping for the determination of center-of-mass accelerations of subsystems (box-11) in accordance with eqn.(3.4.2). The calculation of 'FQDEL' and RMS-values, as per eqn.(3.2.4) and boxes (8,15), is outlined in Table-4. Table 5 illustrates the code used for peak-searching and OLOAD-printing of the RMS-responses of subsystems (box-17).

4.0 INPUT/OUTPUT ILLUSTRATION

As a tutorial example of the application of PSDTOOL, the frame model shown in Figure 2 is analyzed for **RMS-values of random acceleration-response** under uniform **base-excitation**. The frame is defined in the X-Z plane and the excitation consists of two uncorrelated PSD-accelerations along the X and Z directions. The relative-motion approach is chosen for the illustration. The normal modes of the frame are determined in a Phase-I run, with the base-nodes (10 and 30) fully restrained. The Phase-I input and output are standard and are not listed here.

Table 6 presents the input-deck for a restart in Phase-II, consistent with the **'masked-sine'** approach of Section 3.2. The executive deck calls for the PSDTOOL ALTER-packet. Though 'SEMG' is requested in case-control, it is used only for initiating the restart and for processing limited changes in the geometry-input. It does not re-generate the structural matrices. The **OLOAD**-request is non-standard and is used for outputing **all desired RMS-responses in a single table**. The XYPLOT-card asks for PSD-plot under the mask 'RESP' rather than 'PSDF'.

In the bulk-data of Phase-II, all model-definition cards need to be re-entered. The DOF's at grid 99 are assigned to the U1-Set by the USET card so as to provide a reference-base for RMA base-excitation. The TABLED1 cards specify R_{in}, the square-root of the spectral density of excitation, as a function of frequency. For illustration, two subsystems with centers are defined in the DMIG-SECTION and DMIG-DRUGSEC cards. The subsystems are convenient collections of grid-points populating the mass matrix and the centers are convenient reference-points for the subsystems. The DMI-SELDOF cards are optional and select components (1,3,5) for output-presentation.

A large set of parameters, most of which are non-standard, controls the analysis-process. DDRMM=-1 sets up frequency-response solution in matrix-form, so that PSDTOOL can manipulate the solution further. PSDMASK=1 flags the analysis as 'masked-sine' and the required output as RMS-values. DISACC=1 indicates that accelerations of response are desired, rather than displacements . RELMO=1 chooses RMA instead of LMA; CEMMO=1 signals a request for C.M. responses in addition to those at the regular grid-points. GRMSFILT=0.99 helps trap peaks of RMS and RSS accelerations.

In Table 7 is excerpted some of the output obtained from the input-deck of Table 6 for the frame model. Because of the tutorial nature of the illustration (as evidenced, for instance, by the simplistic input-levels and the choice of only four response-frequencies), the numerical magnitudes themselves are irrelevant. For any given subsystem, seven different printouts of RMS-analysis results may appear. Only the output pertaining to subsystem-1 is shown here. Matrix MRUG displays the MR-matrix of eqn.(3.4.1). The so-called 'L O A D' vectors in the next two printouts are the raw and peak-searched values of RMS-accelerations (Q_{out} of eqn.(3.2.4)). The peak-search is done for each of the six components separately. Matrix ACMGRMS denotes the RMS-accelerations at the 'center' (grid 101) of subsystem-1. The last three printouts present RSS-results in the same format.

5.0 CONCLUSION

The Paper has presented the main elements of a DMAP-ALTER framework, called **PSDTOOL**, that significantly enhances the scope of **harmonic/random response** analysis in MSC/NASTRAN. The major benefits that PSDTOOL offers an analyst are :- the automated generation of an OLOAD-style table of grid-point RMS-responses in random analysis; the facility for SORT1/SORT2 peak-search of harmonic/PSD/RMS responses; scanning of RMS-output on the basis of user-defined subsystems; the synthesis of mass-weighted average-responses at 'centers' of subsystems; and, a relative-motion option for analyzing response to base-excitation. These benefits would in turn add to speed and confidence in the design process. The Paper includes DMAP-excerpts and offers a tutorial illustration of input/output features. The application of PSDTOOL to real-life aerospace systems and its extension to element-related output will form the subject of future work.

6.0 ACKNOWLEDGEMENTS

The authors thank I. De Bernardo, M. Wong, T. Tucker and R. McDonald of Spar Aerospace Limited for their encouragement and support.

7.0 REFERENCES

[1] The NASTRAN Theoretical Manual, The MacNeal-Schwendler Corporation, Los Angeles, U.S.A., December 1972.

[2] Clough, R.W. and Penzien, J., Dynamics of Structures, McGraw-Hill, 1975.

[3] MSC/NASTRAN User's Manual, Version 66A, The MacNeal-Schwendler Corporation, Los Angeles, U.S.A., November 1989.



Figure 1: PSDTOOL Overview



Figure 1(Cont'd.): PSDTOOL Overview



Figure 2: Frame Model for Illustration

```
$ DMAP-EXCERPT FOR GENERATION OF RMA EFFECTIVE-LOADS
$
ALTER 1135 $ AFTER FRLG; BEFORE FRRD2.
$ CALL GP4 TO SET UP USETNEW.
GP4      CASES,GEOM4S,EQEXINS,SILS,GPDTS,BGPDTS,CSTMS/
         RMGNEW,YSBNEW,USETNEW,/
         LUSETS/S,N,NOMSET/0/S,N,NOSSET/S,N,NOOSET/
         S,N,NORSET/0/0/S,N,NOSET/S,N,NOLSET/S,N,NOA/0 $
PURGE    RMGNEW,YSBNEW/ALWAYS $
$
$ COMPUTE RIGID-BODY D-MATRIX 'DRBT'.
$ DRBT IS IN 'GLOBAL', DUE TO ORIGIN-MOTION IN BASIC-SYSTEM.
VECPLOT, ,BGPDTS,EQEXINS,CSTMS,,/DRB///4 $
TRNSP    DRB/DRBT/ $
$
$ REVISE DRBT FOR U1-SET. (U1: DISJOINT BASE-PT.)
UPARTN   USETNEW,DRBT/DRBTU1,,,/'G'/'U1'/'S'/1 $
DIAGONAL DRBTU1/IDU1/'SQUARE'/0.0 $ IDENTITY-MATRIX
SOLVE    DRBTU1,IDU1/COEF1/ $
MPYAD    DRBT,COEF1,/DBASE/ $
$
$ CREATE NEW LOAD-MATRICES.
UPARTN   USETNEW,MGG/MU11,,,MZZ/'G'/'U1'/'S' $
UMERGE1  USETNEW,IDU1,,,MZZ/MGGREL/'G'/'U1'/'S' $
UPARTN   USETNEW,PP/PPU1,,,/'G'/'U1'/'S'/1 $
MPYAD    DBASE,PPU1,/DBPP/ $
MPYAD    MGGREL,DBPP,/PPNEW//-1 $
$
EQUIV    PPNEW,PP/ALWAYS $
SSG2     USETD,GMD,,,GOD,,PPNEW/,,PSNEW,PDNEW,/ $
EQUIV    PSNEW,PSSD/ALWAYS $
EQUIV    PDNEW,PDF/ALWAYS $
$
$ IN THE RELMO-METHOD, MPYAD FOR PHFNEW MAY NEED
$ METHOD-SELECTION (PARAM,MPYFLAG), TO SAVE TIME.
$
PARAM    //'STSR'/S,N,MPYDEF/66 $
PARAM    //'STSR'/S,Y,MPYFLAG=0/-66 $
MPYAD    PHIDH,PDF,/PHFNEW/1 $
EQUIV    PHFNEW,PHF/ALWAYS $
PARAM    //'STSR'/MPYDEF/-66 $
$ END OF DMAP-EXCERPT FOR RMA EFFECTIVE-LOADS.
```

Table 1 : DMAP for RMA Effective Loads

```
$ DMAP-EXCERPT FOR DECODING OLB1 INTO MATRIX-FORM
$
ALTER 1297 $ V66A : BEFORE SDR2/SDRXD.
PARAML    OLB1//'TRAILER'/1/S,N,NOLB1 $
PARAML    UGVS//'TRAILER'/1/S,N,NUGVS $
PARAM     //'DIV'/S,N,NSUBC/NUGVS/NOLB1 $
MATGEN    ,/UNIT/1/1 $ 1X1 MATRIX FOR FQT/FQQT ACCUMN.
$
$ FREQ-SET FOR ONE SUBCASE
NNLP = 1 $
DO WHILE (NNLP<=NOLB1) $
PARAM     //'ADD'/S,N,NNLQ/NNLP/2 $ GO FROM WORD-3.
PARAML    OLB1//'DTI'/0/NNLQ/S,N,GYFAC $
CGY = CMPLX(GYFAC) $
ADD       UNIT,/UNITQ/CGY $
APPEND    UNITQ,/FQT/2 $
NNLP = NNLP + 1 $
ENDDO $
TRNSP     FQT/FQ/ $
$
$ EXPAND FQ, FOR A SET OF SUBCASES.
MATGEN    ,/VOB/6/NOLB1/0/NOLB1 $
MATGEN    ,/MER/6/2/1/1 $
NNLP = 1 $
DO WHILE (NNLP<=NSUBC) $
APPEND    FQT,/FQQT/2 $
CGY = CMPLX(NNLP) $
ADD       VOB,/VSUB1/CGY $
TRNSP     VSUB1/VSUB1T/ $
APPEND    VSUB1T,/VSUBT/2 $
$
$ ACCOUNT FOR DMI-SELSUB, IF ANY.
IF(FSELSUB>=0) THEN $
SCALAR    SELSUB//NNLP/1/S,N,MYSUBC $
ADD       VOB,/VOBSU/MYSUBC $
TRNSP     VOBSU/VOBSUT/ $
APPEND    VOBSUT,/VOBSUTT/2 $
ENDIF $
NNLP = NNLP + 1 $
ENDDO $
TRNSP     FQQT/FQQ/ $
MATMOD    FQQ,,,,,/FQQD,/28 $
MERGE     VSUBT,FQQT,,,,MER/SUBFQT/1 $
MATPRT    SUBFQT// $
$ END OF DMAP-EXCERPT FOR DECODING OLB1.
```

Table 2 : DMAP for Decoding Response-Frequencies

```
$ DMAP-EXCERPT FOR MOTION OF CENTER-OF-MASS OF SUBSYSTEM
$
IF (CEMMO>=0) THEN $
$ GENERATE RAW (COMPLEX) ACCNS FOR WHOLE STRUCTURE.
$ WQQD IS A MATRIX OF FREQ-FACTORS. (EQN. 3.1.1)
MPYAD      UGVS,WQQD,/ACCNMO/ $ RAW (COMPLEX) ACCNS.
$
$ LOOP OVER SECTION (SUBSYSTEM).
ISEC = 1 $
DO WHILE  (ISEC<=NSEC) $
$
MESSAGE   //'  SECTION ISEC = '/ISEC $
$
$ REVISE DRBT FOR REF-GRID AT CENTER OF CURRENT SECTION.
$ DRSEC IDENTIFIES (THE 6 DOF'S OF) CENTERS OF SECTIONS.
DELETE     /DVSEC,DRBTU2,IDU2,COEF2,DRUG/ $ NOT-GENERATED.
MATMOD     DRSEC,,,,,/DVSEC,/1/ISEC $
PARTN      DRBT,,DVSEC/,DRBTU2,,/1 $
DIAGONAL   DRBTU2/IDU2/'SQUARE'/0.0 $ IDENTITY-MATRIX
SOLVE      DRBTU2,IDU2/COEF2/ $
MPYAD      DRBT,COEF2,/DRUG/ $
$
$ ISOLATE DRUG AND ACCNMO TO GRIDS ON CURRENT SECTION.
$ MATSEC IDENTIFIES (THE 6 DOF'S OF) SUBSYSTEM-GRIDS.
DELETE     /VECSEC,VECSECD,DROGSEC,ACCSEC,/ $
MATMOD     MATSEC,,,,,/VECSEC,/1/ISEC $
MATMOD     VECSEC,,,,,/VECSECD,/28 $ DIAGONAL
MPYAD      VECSECD,DRUG,/DROGSEC/ $
MPYAD      VECSECD,ACCNMO,/ACCSEC/ $
$
$ COMPUTE 6X6 C.M. MASS-MATRIX MRUG AND ITS INVERSE.
$ EQN. 3.4.1 : MRUG = [D]' [M] [D] ;  MIRUG = MRUG**(-1)
DELETE     /DTMGG,MRUG,MIRUG,,/ $
MPYAD      DROGSEC,MGGCMO,/DTMGG/1 $ TRANSPOSE
MPYAD      DTMGG,DROGSEC,/MRUG/ $
MESSAGE    //'6X6 MASS-MATRIX OF SECTION ISEC= '/ISEC $
MATPRT     MRUG// $
MESSAGE    //' ' $
$ CHECK FOR NULL COLUMNS IN MRUG, BEFORE INVERTING.
DELETE     /VNULL,MRUGN,MIRUGN,IRUGN,/ $
MATMOD     MRUG,,,,,/VNULL,/12/S,N,NULLS/1 $
```

(Contd.)

Table 3 : DMAP-Excerpt for Center-of-Mass Motions

```
$ DMAP FOR C.M. MOTIONS - CONTINUED
$
$ IF MRUG IS ENTIRELY NULL, SECTION HAS NO MASS; SKIP CALCS.
IF(NULLS=6) JUMP LALLNUL $
EQUIV     MRUG,MRUGN/NULLS $
COND      LMUL1,NULLS $
MESSAGE   //'** NO. OF NULL-COLUMNS IN 6X6 MRUG-MATRIX ** ' $
PRTPARM   //0/'NULLS' $
PARTN     MRUG,VNULL,/MRUGN,,,/ $
LABEL     LMUL1 $
DIAGONAL  MRUGN/IRUGN/'SQUARE'/0.0 $ IDENTITY MATRIX
SOLVE     MRUGN,IRUGN/MIRUGN/ $
$ MERGE NULL COLUMNS BACK, IF THEY EXISTED.
EQUIV     MIRUGN,MIRUG/NULLS $
COND      LMUL2,NULLS $
MERGE     MIRUGN,,,,VNULL,/MIRUG/ $
LABEL     LMUL2 $
$
$ EQN. (3.4.2) : COMPUTE CENTRE-OF-MASS ACCELERATIONS.
DELETE    /FRUG,ACCRUG,,,/ $
MPYAD     DTMGG,ACCSEC,/FRUG/ $
MPYAD     MIRUG,FRUG,/ACCRUG/ $
$
$ REVISE ACCNMO TO INCLUDE ACCRUG AT C.M.
DELETE    /ACCNEW,ACC2CO,,,/ $
PARTN     ACCNMO,,DVSEC/ACC2CO,,,/1 $
MERGE     ACC2CO,ACCRUG,,,,DVSEC/ACCNEW/1 $
EQUIV     ACCNEW,ACCNMO/ALWAYS $
$
LABEL     LALLNUL $ C.M. MOTION STAYS AS ZERO.
ISEC = ISEC+1 $
ENDDO $ END OF ISEC-LOOP.
$
$ A NEW ACCNMO-MATRIX RESULTS FROM THE ABOVE ISEC-LOOPING.
$ REVISE UGVS USING NEW ACCNMO THAT INCLUDES ALL CM-ACCRUG'S.
MPYAD     ACCNMO,WQQDR,/UGVSNEW/ $
EQUIV     UGVSNEW,UGVS/ALWAYS $
$
ENDIF $ CEMMO-CONTROL
$ END OF DMAP-EXCERPT FOR CENTER-OF-MASS MOTIONS.
```

Table 3 (Contd.) : DMAP-Excerpt for Center-of-Mass Motions

```
$ DMAP-EXCERPT FOR RMS-CALCULATION : EQN.(3.2.4)
$
$ SET UP FQDEL-VECTOR (FROM FQ) FOR RMS CALCN.
IF (GRMSTYPE <0) THEN $ PSEUDO-RMS
 MATGEN    ,/FQDEL/6/NOLB1/0/NOLB1 $
ELSE $ TRUE-RMS
 NNLP = NOLB1-1 $
 MATGEN    ,/FQU1/6/NOLB1/1/NNLP $
 MATGEN    ,/FQU2/6/NOLB1/NNLP/1 $
 PARTN     FQ,,FQU1/FQV11,FQV12,,/1 $
 PARTN     FQ,,FQU2/FQV21,FQV22,,/1 $
 MERGE     FQV11,FQV21,,,,FQU1/FQW1/1 $
 MERGE     FQV12,FQV22,,,,FQU2/FQW2/1 $
 ADD       FQW2,FQW1/FQDEL/(0.5,0.)/(-0.5,0.) $
 DELETE    /FQU1,FQU2,FQW1,FQW2,/ $
 DELETE    /FQV11,FQV12,FQV21,FQV22,/ $
ENDIF $ GRMSTYPE-CONTROL
$
$ ACCSQ BELOW IS S-OUT OF EQN.(3.2.2).
$ SUM ACCSQ TO SINGLE SUBCASE, FOR DIAGONAL-PSD INPUT.
$
IF(NSUBC>1) THEN $
 NNLP = NUGVS-NOLB1 $
 MATGEN    ,/PVZ/6/NUGVS/NOLB1/NNLP $
 PARTN     ACCSQ,PVZ,/ACSUB1,,,/1 $
 ADD       ACSUB1,/AMAGSV/(0.,0.) $
$
 KPEY=1 $ LOOP-COUNTER
 DO WHILE (KPEY <= NSUBC) $
  DELETE    /PVZ,ACSUB1,AMAGSW,,/ $
  NNLP = (KPEY-1)* NOLB1 $
  NUBY = NUGVS - (NNLP+NOLB1) $
  MATGEN    ,/PVZ/6/NUGVS/NNLP/NOLB1/NUBY $
  PARTN     ACCSQ,PVZ,/,,ACSUB1,/1 $
  ADD       AMAGSV,ACSUB1/AMAGSW/ $
  EQUIV     AMAGSW,AMAGSV/ALWAYS $
  KPEY = KPEY+1 $
 ENDDO $ KPEY-LOOP
$
 ADD       AMAGSV,/AMAGSVF/ $
 EQUIV     AMAGSVF,ACCSQ/ALWAYS $
 NSUBC = 1 $ RESET FOR DIAGONAL PSD-MATRIX.
ENDIF $ NSUBC CONTROL
```

(Contd.)

Table 4 : DMAP-Excerpt for RMS-Calculation

```
$ DMAP for RMS-Calculation (Contd.)
$
$ EQN.(3.2.4) : RMS IS DONE FOR ONLY ONE SARTYPE-LOOP.
$ GYMAGN BELOW IS A PER-DOF (OF SIX) VERSION OF ACCSQ.
IF (JPEY<=1) THEN $
IF (SPEY>=0) THEN $ TYPE-1
 MPYAD     GYMAGN,FQDEL,/GRMS1/ $
ELSE $ TYPE-2
 MPYAD     GYMAGN,FQDEL,/GRMS1/1 $ TRANSPOSE-MPY
ENDIF $ SPEY-CONTROL
$
$ SQUARE-ROOT
DIAGONAL  GRMS1/GRMSQRT/'WHOLE'/0.5 $
EQUIV     GRMSQRT,GRMS1/ALWAYS $
$
$ ACCUMULATE GRMS1 (FOR 6 DOF'S).
ADD       AGRMSAL1,GRMS1/GRMSTEM3/ $
EQUIV     GRMSTEM3,AGRMSAL1/ALWAYS $
$
ENDIF $ JPEY-CONTROL
$ END OF DMAP-EXCERPT FOR RMS-CALCULATION.
```

Table 4 (Contd.) : DMAP-Excerpt for RMS-Calculation

```
$ DMAP-EXCERPT FOR PEAK-SEARCH AND OLOAD-PRINT OF RMS-VALUES
$
ISEC = 1 $
DO WHILE  (ISEC<=NSEC) $
MESSAGE   //' FOR RMS-PRINT, SECTION ISEC = '/ISEC $
$
DELETE    /VECSEC,VECSECD,GRMSSEC,GRSSC,/ $
$
$ ISOLATE AGRMSAL1 (RMS-VALUES) & GRSSCAL (RSS-VALUES)
$ TO GRIDS ON CURRENT SECTION.
MATMOD    MATSEC,,,,,/VECSEC,/1/ISEC $
MATMOD    VECSEC,,,,,/VECSECD,/28 $ DIAGONAL
MPYAD     VECSECD,AGRMSAL1,/GRMSSEC/ $
MPYAD     VECSECD,GRSSCAL,/GRSSC/ $
$
IF ((GRUNFILT=0) OR (GRUNFILT=2)) THEN $
DELETE    /OGRMSAL1,,,,/ $
MESSAGE   //'OLOAD-PRINT OF UNFILTERED RMS OF ISEC='/ISEC/
          ' FOLLOWS' $
SDR2      CASEDR,,,,EQEXINS,,,,,
          GRMSSEC,,,,/OGRMSAL1,,,,,/'STATICS' $
OFP       OGRMSAL1,,,,,//S,N,CARDNO $
ENDIF $ GRUNFILT-CONTROL
$
$ MAX-FILTERING OF RMS-VECTOR
$
$ SEPARATE THE RMS-VECTOR INTO SIX DOF-COLUMNS.
DELETE    /AGRMSEXP,AGRMSC6,,,/ $
MPYAD     GRMSSEC,UNIR1,/AGRMSC6/ $ 6 REPEAT-COLS. OF RMS-VECTOR
ADD       IDENC6,AGRMSC6/AGRMSEXP///1 $
EQUIV     AGRMSEXP,GRMSSEC/ALWAYS $
$
$ FILTER THE SIX COLUMNS OF GRMSSEC WITH GRMSFILT.
$ AVOID PROCESSING NULL COLUMNS, IF ANY.
$
DELETE    /VNULL,AGRMNUL,AGRMFUL,,/ $
MATMOD    GRMSSEC,,,,,/VNULL,/12/S,N,NULLS/1 $ NULL-COLS
EQUIV     GRMSSEC,AGRMFUL/NULLS $
COND      LGNU1,NULLS $
MESSAGE   //'** NO. OF NULL-COLUMNS IN 6-COL RMS ** ' $
PRTPARM   //0/'NULLS' $
```

(Contd.)

Table 5 : DMAP-Excerpt for Search/Print of RMS

```
$ DMAP-Excerpt for Search/Print of RMS (Contd.)
$
PARTN     GRMSSEC,VNULL,/AGRMFUL,,AGRMNUL,/1 $
LABEL     LGNU1 $
DELETE    /AGRMX,AGRMXD,AGRMXR,AGRNORX,/ $
DELETE    /AGRMFILX,AGRMFAX,AGRMSMAX,,/ $
MATMOD    AGRMFUL,,,,,/AGRMX,/7 $ COL-MAX
MATMOD    AGRMX,,,,,/AGRMXD,/28 $
DIAGONAL  AGRMXD/AGRMXR/SQUARE/-1.0 $
MPYAD     AGRMFUL,AGRMXR,/AGRNORX/ $
MATMOD    AGRNORX,,,,,/AGRMFILX,/2////GRMSFILT $
MPYAD     AGRMFILX,AGRMXD,/AGRMFAX/ $
EQUIV     AGRMFAX,AGRMSMAX/NULLS $
COND      LGNU2,NULLS $
MERGE     AGRMFAX,,AGRMNUL,,VNULL,/AGRMSMAX/1 $
LABEL     LGNU2 $
MESSAGE   //'OLOAD-PRINT OF MAX-FILTERED RMS OF ISEC='/
          ISEC/'  FOLLOWS : ' $
MESSAGE   //'CAUTION :- ZEROS MAY BE TRUE OR DUE TO FILTER.' $
DELETE    /AGRMSMX2,OGRMSMX1,,,/ $
$ COMBINE THE SIX COLUMNS TOGETHER INTO ONE COLUMN.
MPYAD     AGRMSMAX,UNIC1,/AGRMSMX2/ $
EQUIV     AGRMSMX2,AGRMSMAX/ALWAYS $
SDR2      CASEDR,,,,EQEXINS,,,,,
        AGRMSMAX,,,,/OGRMSMX1,,,,,/'STATICS' $
OFP       OGRMSMX1,,,,,//S,N,CARDNO $
$
$ MATGPR-PRINT OF CENTRE-OF-MASS RMS
IF(CEMMO>=0) THEN $
DELETE    /DVSEC,DVSECD,ACMGRMS,GCMRSS,/ $
MESSAGE   //'MATGPR-PRINT OF RMS AT CENTRE-OF-MASS OF ISEC='/
          ISEC/'  FOLLOWS' $
MATMOD    DRSEC,,,,,/DVSEC,/1/ISEC $
MATMOD    DVSEC,,,,,/DVSECD,/28 $ DIAGONAL
MPYAD     DVSECD,AGRMSAL1,/ACMGRMS/ $
MATGPR    GPLS,USET,SILS,ACMGRMS//'H'/'G' $ ONE COLUMN
ENDIF $ CEMMO-CONTROL OF C.M. RMS-PRINT
$
$ (ETC.)   (ETC.)   (ETC.)  -- FOR RSS --
$
ISEC = ISEC+1 $
ENDDO $ END OF ISEC-LOOP FOR RMS-PRINT
$ END OF DMAP FOR PEAK-SEARCH AND OLOAD-PRINT OF RMS-VALUES.
```

Table 5 (Contd.) : DMAP-Excerpt for Search/Print of RMS

```
INIT MASTER(S)
ASSIGN USERFILE='TEMP_RMA.F49',STATUS=UNKNOWN,
      UNIT=49,FORM=FORMATTED ,TEMP $ CORSYS DUMP.
ASSIGN MODES='MODES.MASTER' $ V66A
DBLOCATE DB=*, VERSION=1, LOGI=MODES
DBLOCATE PARM=*, VERSION=1, LOGI=MODES
$-----------------------------------------------------------------
ID GRMS,PSDTOOL
TIME 2000
DIAG 8,64
SOL 71 $ MODAL FREQ. RESP. V66A
$-----------------------------------------------------------------
COMPILER NOLIST,NOREF,NODECK
$ INSERT PSDTOOL-ALTER.
INCLUDE 'PSDTOOL.ALT'
$-----------------------------------------------------------------
CEND
TITLE = V66A FRAME-EXAMPLE : BASE-MOTION ; NO DAMPING
SUBTI = SOL-71 DBLOC-RST FROM SOL-63; REL-MOTION METHOD
LABEL = 2-PSD : S(X)=1.0 ; S(Z)=0.25; GRMS-RUN; 2 SECTS/CENTS
ECHO  = SORT
$-----------------------------------------------------------------
SEMG  = ALL $ COSMETIC ONLY.
SELG = ALL $ LOAD-GENERATION IS REQD.
SELR = ALL $ LOAD-REDUCTION
$
$ FULL GRMS-PRINT VIA OLOAD
OLOAD(SORT1) = ALL
$
$ TOKEN OR TRUE DISP/VELO/ACCEL REQUEST IS MANDATORY.
SET 17 = 99 $ TOKEN ONLY.
ACCEL(SORT1,PHASE,PLOT) = 17
$-----------------------------------------------------------------
SPC = 33
$SDAMP  = 100
FREQ = 100 $
$
$ TWO UNCORRELATED RANDOM-INPUTS
LOADSET = 1000
SUBCASE 1
 DLOAD = 101 $
SUBCASE 2
 DLOAD = 102 $
```

(Contd.)

Table 6 : Sample Input for Phase-II of Frame Example

```
OUTPUT(XYPLOT) $ PSD-PLOT VIA MASKED 'ACCE-SINE'
XYPRINT,XYPLOT ACCE RESP /22(T1RM),22(T3RM) $
BEGIN BULK
$-------------------------------------------------------------------
$
$ REPEAT MODE-EXTRACTION BULK-DECK FOR PARTIAL-SEMG.
$
$ BASIC MODEL DEFINITION
INCLUDE 'BASIC_MODEL.NAS'
$
$ CONSTRAINTS ON OUT-OF-PLANE MOTION
SPC1,33,246,11,31
SPC1,33,246,20,THRU,24
$ CLAMPED BASE FOR RELMO.
SPC1,33,123456,10,30
PARAM,AUTOSPC,YES
$
$ BASE-GRID FOR EXCITATION (DISJOINT)
GRID,99,,60.,0.,0. $
$
$ (UNUSED) LMA BASE-MODIFICATION
$RBE2,99,99,123456,10,30 $
$CONM2,999,99,,1.E5
$SPC1,33,2456,99
$SUPORT,99,13
$
$ POSSIBLE CENTERS OF SUBSYSTEMS
$ LUMPED-MASS; STRADDLING T-PLT CAUSES NO MASS-COUPLING.
GRID,100,,60.0,0.0,57.53424 $ WHOLE-MODEL CENTER
GRID,101,,60.0,0.0,40.0 $ 2-COLUMNS CENTER (APPROX.)
GRID,102,,60.0,0.0,80.0 $ HORIZ. BEAM CENTER (APPROX.)
$
$-------------------------------------------------------------------
$ (UNUSED) DAMPING
TABDMP1,100
,0.001,0.04,10000.,0.04,ENDT
$
$ SOLUTION-FREQUENCIES
$ THIS SMALL SET IS FOR ILLUSTRN. ONLY; GRMS NOT ACCURATE.
FREQ,100,8.7,22.5,31.3,60.9
$-------------------------------------------------------------------
```

(Contd.)

Table 6(Contd.) : Sample Input for Phase-II of Frame Example

```
$ MASKED-SINE (RMA) BASE-EXCITATION
$
$ DEFINE U1-SET (RMA)
USET,U1,99,123456
$
$ CASE-101 : X-DIRECTION
DLOAD,101,1.0,1.0,300 $
RLOAD2,300,300,,,9999
LSEQ,1000,300,1010
$FORCE,1010,99,0,1.E5,1.,0.,0. $ LMA
FORCE,1010,99,0,1.,1.,0.,0.
TABLED1,9999
,0.,1.,10000.,1.,ENDT $ (STANDARD PSD=1.0 TOO.)
$
$ CASE-102 : Z-DIRECTION
DLOAD,102,1.0,1.0,350 $
RLOAD2,350,350,,,9222
LSEQ,1000,350,1050
$FORCE,1050,99,0,1.E5,0.,0.,1. $ LMA
FORCE,1050,99,0,1.,0.,0.,1.
TABLED1,9222
,0.,.5,10000.,.5,ENDT $ (STANDARD PSD=0.25.)
$------------------------------------------------------------
$
$ DEFINITION OF TWO SUBSYSTEMS
$
$ INCLUSION OF C.M. IN DMIG-SECTION IS OPTIONAL.
$ HEADER-CARD FOR SECTION-GRIDS
DMIG,SECTION,0,9,1,,,,20
$
$ DEFINE SECTION-1 GRIDS (ON COLUMNS)
DMIG,SECTION,1,,,10,1,1.0
,11,1,1.0,,12,1,1.0
,30,1,1.0,,31,1,1.0
,32,1,1.0
,101,1,1.0 $
$ DEFINE SECTION-2 GRIDS (ON HORIZ. BEAM)
DMIG,SECTION,2,,,20,1,1.0
,21,1,1.0,,22,1,1.0
,23,1,1.0,,24,1,1.0
,102,1,1.0 $
```

(Contd.)

Table 6(Contd.) : Sample Input for Phase-II of Frame Example

```
$ DEFINITION OF SUBSYSTEM-CENTERS (I.E., REF. POINTS)
$
$ HEADER CARD
DMIG,DRUGSEC,0,9,1,,,,20
$ DEFINE SECTION-1 CENTER
DMIG,DRUGSEC,1,,,101,1,1.0
$ DEFINE SECTION-2 CENTER
DMIG,DRUGSEC,2,,,102,1,1.0
$-----------------------------------------------------------
$ OPTIONAL SUBC-SELECTION; DEF = ALL (SUBOPT CONTROLS HOW.)
$DMI,SELSUB,0,2,1,1,,50,1 $ SAFE UPPER-BOUND FOR 50 SUBCASES.
$DMI,SELSUB,1,1,1.,2,1.
$
$ OPTIONAL OUTPUT DOF-SELECTION; DEFAULT = ALL SIX.
DMI,SELDOF,0,2,1,1,,6,1 $ HEADER
DMI,SELDOF,1,1,1.,3,1.,5,1.
$-----------------------------------------------------------
$
$ PROCESS PARAMETERS
$
$PARAM,LFREQ,0.
PARAM,HFREQ,200.
PARAM,DDRMM,-1        $ MANDATORY FOR SEARCH ETC.
$
$PARAM,CORSYS,15      $ OUTPUT CO-ORD SYSTEM (OLD OR NEW)
$PARAM,CORFILE,49     $ DUMP-FILE FOR CORSYS.
$
PARAM,RELMO,1         $ WANT RMA (RATHER THAN LMA).
PARAM,MOTOT,1         $ IN RMA, WANT TOTAL-MOTION.
PARAM,CEMMO,1         $ WANT C.M. MOTION.
$
$PARAM,PRTSEC,1       $ PRINT SECTION-INFO.
PARAM,MPYFLAG,243     $ MPYAD METHOD=2 (UDV; PHFNEW).
PARAM,PSDMASK,1       $ MASKED-SINE; ONLY RMS-PROCESSING
PARAM,SUBOPT,0        $ GRMS IMPLIES ONE SUBC-LOOP ONLY.
PARAM,GRMSTYPE,1      $ 'TRUE' GRMS; OLOAD-PRINT
PARAM,MAXFILT,0.99    $ FILTER FOR SINE/PSD
PARAM,GRMSFILT,0.99$ FILTER FOR GRMS
PARAM,GRUNFILT,2      $ PRINT UNFILTERED GRMS & GRSS ALSO.
PARAM,DISACC,1        $ WANT ACCEL (RATHER THAN DISP).
PARAM,SARTYPE,0       $ SORT1 TYPE OF MAX-SEARCH
$PARAM,PARTONE,1      $ PARTN-OVERRIDE FOR SORT1 SEARCH
$
ENDDATA
```

Table 6(Contd.) : Sample Input for Phase-II of Frame Example

```
^^^6X6 MASS-MATRIX OF SECTION ISEC=            1
                                             INTERMEDIATE MATRIX ... MRUG

     3.772000E-03    0.000000E+00    0.000000E+00    0.000000E+00    0.000000E+00    0.000000E+00
     0.000000E+00    3.772000E-03    0.000000E+00    0.000000E+00    0.000000E+00   -8.673617E-19
     0.000000E+00    0.000000E+00    3.772000E-03    0.000000E+00    8.673617E-19    0.000000E+00
     0.000000E+00    0.000000E+00    0.000000E+00    2.944000E+00    0.000000E+00    0.000000E+00
     0.000000E+00    0.000000E+00    8.673617E-19    0.000000E+00    1.652320E+01    0.000000E+00
     0.000000E+00   -8.673617E-19    0.000000E+00    0.000000E+00    0.000000E+00    1.357920E+01

^^^=====================================================
^^^OLOAD-PRINT OF UNFILTERED GRMS OF ISEC=          1   FOLLOWS
^^^=====================================================
                                              L O A D   V E C T O R

     POINT ID.    TYPE      T1              T2              T3             R1              R2             R3
          10       G     7.224957E+00    0.0          3.612478E+00    0.0           0.0            0.0
          11       G     4.648129E+03    0.0          6.100815E+00    0.0           4.916500E+01   0.0
          12       G     2.071083E+03    0.0          8.687554E+00    0.0           1.652364E+02   0.0
          30       G     7.224957E+00    0.0          3.612478E+00    0.0           0.0            0.0
          31       G     4.648129E+03    0.0          6.100815E+00    0.0           4.916500E+01   0.0
          32       G     2.071083E+03    0.0          8.687554E+00    0.0           1.652364E+02   0.0
         101       G     1.453517E+03    0.0          5.920788E+00    0.0           4.623353E+00   0.0

^^^=====================================================
^^^OLOAD-PRINT OF MAX-FILTERED GRMS OF ISEC=          1   FOLLOWS :
^^^CAUTION :- ZEROS MAY BE TRUE OR DUE TO FILTER.
^^^=====================================================
                                              L O A D   V E C T O R

     POINT ID.    TYPE      T1              T2              T3             R1              R2             R3
          11       G     4.648129E+03    0.0          0.0             0.0           0.0            0.0
          12       G     0.0             0.0          8.687554E+00    0.0           1.652364E+02   0.0
          31       G     4.648129E+03    0.0          0.0             0.0           0.0            0.0
          32       G     0.0             0.0          8.687554E+00    0.0           1.652364E+02   0.0

^^^MATGPR-PRINT OF GRMS AT CENTRE-OF-MASS OF ISEC=          1   FOLLOWS
0                   ACMGRMS
0COLUMN        1
        101 T1   1.45352E+03        101 T3   5.92079E+00        101 R2   4.62335E+00

^^^=====================================================
^^^OLOAD-PRINT OF UNFILTERED RT-SUM-SQ GRSS OF ISEC=          1   FOLLOWS :
^^^CAUTION :- GRSS IS ALWAYS PRINTED AS T1-DOF.
^^^=====================================================
                                              L O A D   V E C T O R

     POINT ID.    TYPE      T1              T2              T3             R1              R2             R3
          10       G     8.077747E+00    0.0          0.0             0.0           0.0            0.0
          11       G     4.648133E+03    0.0          0.0             0.0           0.0            0.0
          12       G     2.071101E+03    0.0          0.0             0.0           0.0            0.0
          30       G     8.077747E+00    0.0          0.0             0.0           0.0            0.0
          31       G     4.648133E+03    0.0          0.0             0.0           0.0            0.0
          32       G     2.071101E+03    0.0          0.0             0.0           0.0            0.0
         101       G     1.453529E+03    0.0          0.0             0.0           0.0            0.0

^^^=====================================================
^^^OLOAD-PRINT OF MAX-FILTERED RT-SUM-SQ GRSS OF ISEC=          1   FOLLOWS :
^^^CAUTION : T1-ZEROS MAY BE TRUE OR DUE TO FILTER.
^^^=====================================================
                                              L O A D   V E C T O R

     POINT ID.    TYPE      T1              T2              T3             R1              R2             R3
          11       G     4.648133E+03    0.0          0.0             0.0         - 0.0            0.0
          31       G     4.648133E+03    0.0          0.0             0.0           0.0            0.0

^^^MATGPR-PRINT OF RT-SUM-SQ GRSS AT C.M. OF ISEC=          1   FOLLOWS :
0                   GCMRSS
0COLUMN        1
        101 T1   1.45353E+03
```

Table 7 : Output-Excerpts from Phase-II of Frame Example